DESCRIPTION

This project to show how to monitor the data using NodeMCU with Favoriot platform. This project used ESP32 NodeMCU as a wifi module and data read from DHT22 as a temperature and humidity sensor. Favoriot as a middleware platform that can show the realtime data on dashboard.

COMPONENTS


Female To Female Jumper (3)
DHT22 Sensor
USB MicroB Type Cable

SCHEMATIC DIAGRAM

1. Construct the circuit as shown in diagram below:



SETUP ARDUINO IDE

1. Connect ESP32 board using USB MicroB cable to your USB port on your computer.
2. Open the Arduino IDE software > File > Preferences > Additional Board Manager URLs > https://dl.espressif.com/dl/package_esp32_index.json > Click OK



3. At Arduino IDE software > Tools > Board > Boards Manager > Search ESP32, then installed



4. At Arduino IDE software > Tools > Board > ESP32 Dev Module



5. Then, write the completed code.

WRITE THE CODE

This is a combine code from Arduino program and Favoriot platform. The data will be retrieved from the sensor and will be displayed on the Favoriot platform in realtime. Here, the completed code for Arduino code.

```
#include <WiFi.h>
#include <HTTPClient.h>
#include <ArduinoJson.h>
#include <SimpleDHT.h>

#define APIKEY  "Your API key"                          //replace API key
#define DEVICE_DEV_ID "Your device developer ID"        //replace device developer ID

const int SEN_PIN = 19;
SimpleDHT22 dht22(SEN_PIN);

const char ssid[] = "Your wifi SSID";                   //replace wifi SSID
const char password[] = "Your Wifi Password";           //replace wifi password

float temperature = 0;
float humidity = 0;
long previousMillis = 0;
int interval = 5000; // 5 seconds

void setup()
{
  Serial.begin(115200);

  WiFi.mode(WIFI_STA);
  Serial.print("Connecting to ");
  Serial.print(ssid);
  WiFi.begin(ssid, password);
  while (WiFi.status() != WL_CONNECTED) {
          delay(500);
          Serial.print(".");
  }

  Serial.println("Connected!");
  Serial.print("IP Address: ");
  Serial.println(WiFi.localIP());
}

void loop()
{
```

```
 if (millis() - previousMillis > interval) {
        int err = SimpleDHTErrSuccess;
        if ((err = dht22.read2(&temperature, &humidity, NULL)) != SimpleDHTErrSuccess) {
            Serial.print("Read DHT22 failed, err="); Serial.println(err);
            delay(1000);
            return;
        }

  Serial.print("Sample OK: ");
  Serial.print((float)temperature);
  Serial.print(" *C, ");
  Serial.print((float)humidity);
  Serial.println(" RH%");

  StaticJsonDocument<200> doc;

  JsonObject root = doc.to<JsonObject>();                    // Json Object refer to { }
  root["device_developer_id"] = DEVICE_DEV_ID;

  JsonObject data = root.createNestedObject("data");
  data["temperature"] = (int)temperature;
  data["humidity"] = (int)humidity;

  String body;
  serializeJson(root, body);
  Serial.println(body);

  HTTPClient http;

  http.begin("http://apiv2.favoriot.com/v2/streams");
  http.addHeader("Content-Type", "application/json");
  http.addHeader("Apikey", APIKEY);

  int httpCode = http.POST(body);
  if (httpCode > 0) {
    Serial.printf("[HTTP] POST... code: %d\n", httpCode);
    if (httpCode == HTTP_CODE_OK) {
      String payload = http.getString();
      Serial.println(payload);
    }
  }
  else {
    Serial.printf("[HTTP] POST... failed, error: %s\n", http.errorToString(httpCode).c_str());
  }
  http.end();
  previousMillis = millis();
 }
}
```

RESULT

Once finished, click on the Verify button and Upload code to board. The output will be display on Serial Monitor.



Open Favoriot platform at browser, click at Data Stream and select a device. The output will be displayed as shown below.



To display in graph, select Graphboards > Create Graphboard > Completed all required